IMPLEMENTING AN AFFINE CIPHER SYSTEM

The code below is a computerised affine cipher system written in MicroWorlds Pro. The program creates its own interface from scratch – there is no need to create buttons, text boxes, etc and set their properties; simply enter the text below into the procedures area of MW Pro, and then type startup into the Command Centre.
The danger with this approach is that if the program is saved with the interface and all its objects, the program will generate error messages the next time it is run because the interface objects will already exist and the program will be unable to create them. This is the reason for the exit button – to remove the interface objects before saving the program.
A set of notes describing the operation of affine ciphers has been included at the end of this document.
=========================
AFFINE CIPHER SYSTEM 

=========================
David Dimsey 2005
=============================

A plaintext message is entered into the "plaintext" box. The sliders are used to set the multiplying factor and the added constant for the affine cipher:

C = aP + b

where P is the plaintext and C is the ciphertext.

Note that an affine cipher with multiplier of 1 is just a simple shift (or "Caesar") cipher, while an affine cipher with shift of 0 is a simple multiplication cipher.

=============================

MicroWorlds Pro v1.07

=============================
to startup

create_interface

make "characters [a b c d e f g h i j k l m n o p q r s t u v w x y z A B C D E F G H I J K L M N O P Q R S T U V W X Y Z]

make "coprimes [1 3 5 7 9 11 15 17 19 21 23 25]

make "inverses [1 9 21 15 3 19 7 23 11 5 17 25]
end

to create_interface

newbutton "clearbutton [-65 -95][clearall]

set "clearbutton "size [130 25]

newtext "text1 [-360 205][300 36]

settext1 [The Affine Cipher]

text1, top select bottom

setfont "Arial

setfontsize 18

settc 19

transparent "text1

newtext "text2 [10 200][300 36]

settext2 [Please click the 'exit' button before closing this program]

text2, top select bottom

setfont "Arial

setfontsize 8

settc 19

transparent "text2

newbutton "exitbutton [305 197][exit]

set "exitbutton "size [50 20]

newtext "plaintext [-360 160][350 250]

newtext "ciphertext [10 160][350 250]

newbutton "clearplainbutton [-280 -120][clear plainbox]

set "clearplainbutton "size [130 25]

newbutton "clearcipherbutton [150 -120][clear cipherbox]

set "clearcipherbutton "size [130 25]

newbutton "encryptbutton [-280 -150][encrypt plaintext]

set "encryptbutton "size [130 40]

newbutton "decryptbutton [150 -150][decrypt ciphertext]

set "decryptbutton "size [130 40]

newslider "shift [-125 -150][0 25 0]

newslider "multiplier [10 -150][1 26 1]

setbg 40

end

to exit

remove "clearbutton

remove "text1

remove "plaintext

remove "ciphertext

remove "clearcipherbutton

remove "clearplainbutton

remove "encryptbutton

remove "decryptbutton

remove "text2

remove "shift

remove "multiplier

remove "exitbutton

end

to encrypt :plaintext

if not member? multiplier :coprimes [announce [Not a valid multiplier] setmultiplier 1 stop]

make "ptext caps checktext :plaintext

make "ctext "
encrypt1 :ptext

setciphertext fives :ctext

ciphertext, top select bottom 

setfont "|Courier New| setstyle "regular

top

end

to encrypt1 :ptext

if empty? :ptext [stop]

make "ctext word :ctext char sum mod sum product difference ascii first :ptext 65 multiplier shift 26 65
encrypt1 bf :ptext

end

to decrypt :ciphertext

make "ctext lower checktext :ciphertext

make "ptext "

findpos multiplier :coprimes

make "mult item :spot :inverses

decrypt1 :ctext

setplaintext :ptext

plaintext, top select bottom

setfont "|Courier New| setstyle "regular

top

end

to decrypt1 :ctext

if empty? :ctext [stop]

make "current difference ascii first :ctext 97

make "ptext word :ptext char sum mod product difference :current shift :mult 26 97

decrypt1 bf :ctext

end

to clear :boxname

if equal? :boxname "cipherbox [setciphertext "]

if equal? :boxname "plainbox [setplaintext "]
end

to findpos :num :list

if empty? :list [stop]

make "spot 1

findpos1 :num :list

end

to findpos1 :num :list

if equal? :num first :list [stop]

make "spot sum :spot 1

findpos1 :num bf :list

end
to cipherbox

op "cipherbox

end

to plainbox

op "plainbox

end

to clearall

setplaintext "

setciphertext "

end

to checktext :intext

if empty? :intext [op " ]

if member? first :intext :characters 


[op word first :intext checktext bf :intext ]

checktext bf :intext

end

to lower :inword

if equal? :inword " [op " ]

ifelse (ascii first :inword) > 96 


[op word first :inword lower bf :inword]

[op word char ((ascii first :inword) + 32) lower bf :inword]

end

to caps :inword

if equal? :inword " [op " ]

ifelse (ascii first :inword) > 96 


[op word char ((ascii first :inword) - 32) caps bf :inword]


[op word first :inword caps bf :inword]

end

to fives :inword

if (count :inword) < 5 [op :inword]

op se word first :inword word first bf :inword word first bf bf :inword word first bf bf bf :inword first bf bf bf bf :inword

 fives bf bf bf bf bf :inword

end

to mod :a :b

if :a > 0 [op remainder :a :b]
op mod sum :a :b :b

end

The program creates this interface:
[image: image1.png]JIK[LIM[N[O[P[Q[R[S|T|U|VIW

A|/B|C|D/E[F|G/H

t n
AL TP KUPAL

m
e

e
LLA




SOME NOTES ABOUT AFFINE CIPHERS

1.
Additive (or Shift) ciphers

The well-known “Caesar” cipher is one example of the class of ciphers known as shift ciphers. In these cipher systems, the algorithm is to represent each letter in the plaintext by the letter that is k places further on in the alphabet (hence k is the key for the cipher). The classic Caesar cipher has a shift of 3. 

To make the process of encryption and decryption easier and more amenable to computer processing, it is best to represent the letters of the alphabet as numbers from 0 to 25, with A ( 0, B ( 1, C ( 2, and so on. 

The generalised shift cipher is one in which the letters of the plaintext can be shifted by adding any number to obtain the cipher text; well, not just any number: a shift of 0 will produce a ciphertext that is identical with the plaintext, with no concealment at all of the message. Similarly, a shift of 26 will also produce a ciphertext that is identical with the plaintext. 

Also, a shift of 27 will be identical with a shift of 1; a shift of 28 identical with a shift of 2, and so on. We therefore need only to consider ciphers with shifts between 1 and 25. 

This is actually modular arithmetic, with the modulus being 26. The letters of the alphabet are associated with the integers modulo 26. These integers are just the possible remainders when an integer is divided by 26.

Our shift cipher is then defined by the rule:


C  =  P  +  k  (mod 26)

Where P is a letter of the plaintext, expressed as a number from 0 to 25, k is the key, and is a whole number in the range 1 to 25, and C is the corresponding ciphertext letter, expressed as a whole number from 0 to 25.

A coding scheme for representing letters as numbers does exist for use in computers, and for other forms of communication. The scheme is the American Standard Code for Information Interchange, or the ASCII code. In this code, A = 65, B = 66, through to Z = 90, and a = 97, b = 98 through to z = 122. 

To implement a shift cipher, the process is:

For each letter of the plaintext

Convert the letter in upper case (if it is not already)

Express this letter as its ASCII code

Subtract 65 from the ASCII code

Add the key value k 

Express the result modulo 26

Add 65

Convert this ASCII code to its corresponding letter

Repeat until all the letters of the plaintext have been encrypted

Notice that for a shift cipher, a letter cannot be encrypted as itself.

ASCII code is used in all computer applications, and an application such as a spreadsheet can be used to construct a very primitive computerised encryption tool that implements this process:

The formula used in cell A4 is:

=IF(A1=” “,” “,CHAR((MOD((CODE(UPPER(A1))-65)+7,26))+65))

The plaintext is entered in row 1 of the spreadsheet, one letter per cell, and the cipher text appears in row 4 of the spreadsheet.

Decryption simply reverses the process:

The formula in cell A7 is:

=IF(A4=” “,” “,CHAR(MOD((CODE(A4)-65)-7,26)+97))

2.
Multiplicative ciphers 
If we can create a cipher based on adding some number (mod 26) to our plaintext, then we can surely create one that multiplies the numerical plaintext values to obtain the ciphertext.

For instance, if we multiplied the plaintext values by, say, 3, we have:

	plaintext
	t
	o
	
	b
	e
	
	o
	r
	
	n
	o
	t
	
	t
	o
	
	b
	e

	plaintext number
	19
	14
	
	1
	4
	
	14
	17
	
	13
	14
	19
	
	19
	14
	
	1
	4

	ciphertext number
	5
	16
	
	3
	12
	
	16
	25
	
	13
	16
	5
	
	5
	16
	
	3
	12

	ciphertext
	F
	Q
	
	D
	M
	
	Q
	Z
	
	N
	Q
	F
	
	F
	Q
	
	D
	M


Notice that “A” (with a numerical value of 0) will always be encrypted as itself in this system.

There is a problem with the multiplicative cipher if we use 4 as the multiplier:

	plaintext
	t
	o
	
	b
	e
	
	o
	r
	
	n
	o
	t
	
	t
	o
	
	b
	e

	plaintext number
	19
	14
	
	1
	4
	
	14
	17
	
	13
	14
	19
	
	19
	14
	
	1
	4

	ciphertext number
	24
	4
	
	4
	16
	
	4
	16
	
	0
	4
	24
	
	24
	4
	
	4
	16

	ciphertext
	Y
	E
	
	E
	Q
	
	E
	Q
	
	A
	E
	Y
	
	Y
	E
	
	E
	Q


OOPS! Both “o” and “b” from the plaintext have been encrypted as “E”, while both “e” and “r” have been encrypted as “Q”. The recipient of the message will not be able to decipher it, because she will not know whether to decipher the ciphertext “Q” as “e” or “r”, nor whether to decipher the letter “E” as “o” or “b”. A cryptosystem in which encrypted messages cannot be accurately decrypted is not very useful.

It seems that we cannot multiply by just any number. We need a multiplier for which all of its multiples mod 26 must be different. This occurs if the multiplier k is relatively prime to the modulus 26. That is, the greatest common divisor of 26 and k is 1.

It is easy to verify that the allowable set of multipliers is {1, 3, 5, 7, 9, 11, 15, 17, 19, 21, 23 25}

Decryption of a multiplicative cipher presents some difficulty. Reversing the multiplication process is not just a matter of dividing the ciphertext number by the mutiplier to obtain the plaintext number. In our first example above, the multiplier was 3. However, not all of the ciphertext numbers are multiples of 3. Hnece, if we just divide all of the ciphertext numbers by 3,we get some non-integer results, which cannot be converted to letters at all. How then do we decrypt a multiplicative cipher? 
With a modular arithmetic such as we have been using, we need to identify that the inverse operation to multiplication is not division, but the operation of multiplying by the multiplicative inverse of the number involved. The multiplicative inverse of a number modulo 26 is the number that must be multiplied by the first number to give a value of 1 (which is called the identity for multiplication). 

To find the multiplicative inverses for the set of allowable multipliers, we need to perform all of the possible multiplications, and find those that give a result of 1.

It is a relatively easy task to set up a spreadsheet to perform this calculation for us: 

	
	 
	1
	3
	5
	7
	9
	11
	15
	17
	19
	21
	23
	25
	

	
	1
	1
	3
	5
	7
	9
	11
	15
	17
	19
	21
	23
	25
	

	
	3
	3
	9
	15
	21
	1
	7
	19
	25
	5
	11
	17
	23
	

	
	5
	5
	15
	25
	9
	19
	3
	23
	7
	17
	1
	11
	21
	

	
	7
	7
	21
	9
	23
	11
	25
	1
	15
	3
	17
	5
	19
	

	
	9
	9
	1
	19
	11
	3
	21
	5
	23
	15
	7
	25
	17
	

	
	11
	11
	7
	3
	25
	21
	17
	9
	5
	1
	23
	19
	15
	

	
	15
	15
	19
	23
	1
	5
	9
	17
	21
	25
	3
	7
	11
	

	
	17
	17
	25
	7
	15
	23
	5
	21
	3
	11
	19
	1
	9
	

	
	19
	19
	5
	17
	3
	15
	1
	25
	11
	23
	9
	21
	7
	

	
	21
	21
	11
	1
	17
	7
	23
	3
	19
	9
	25
	15
	5
	

	
	23
	23
	17
	11
	5
	25
	19
	7
	1
	21
	15
	9
	3
	

	
	25
	25
	23
	21
	19
	17
	15
	11
	9
	7
	5
	3
	1
	


So, for the set of allowable multipliers {1, 3, 5, 7, 9, 11, 15, 17, 19, 21, 23 25}, the corresponding inverses are  {1, 9, 21, 15, 3, 19, 7, 23, 11, 5, 17, 25}
In our first example, the multiplier was 3. To decrypt the message, we multiply each of the ciphertext numbers by 9, the multiplicative inverse of 3:

	ciphertext 
	F
	V
	M
	
	J
	Y
	M
	
	Y
	C
	
	G
	A
	C
	F

	ciphertext number
	5
	21
	12
	
	9
	24
	12
	
	24
	2
	
	6
	0
	2
	5

	plaintext number
	19
	7
	4
	
	3
	8
	4
	
	8
	18
	
	2
	0
	18
	19

	plaintext
	t
	h
	e
	
	d
	i
	e
	
	i
	s
	
	c
	a
	s
	T


5 x 9 = 45 = 26 x 1 + 19 = 19 (mod 26)

3.
Affine ciphers
An affine cipher combines both the additive and multiplicative ciphers by using two keys, say a and b. A plaintext letter is represented in numerical form, and then this number P is encrypted as C according to the equation:

C  =  (aP  +  b) mod 26

The multiplier a must be relatively prime to 26, and the additive key b is one of the integers in the range 0 – 25.

Decryption of an affine cipher reverses this process:

P  =  a-1(C  -  b) mod 26

Where a-1 is the multiplicative inverse of the multiplier a.
For example, in an affine cipher system with a = 11 and b = 8:

Plaintext:
imagination is more important than knowledge

“i” has the numerical equivalent 8, (8 x 11 + 8) mod 26  = 96 (mod 26) = 18 = “S”

“m” has the numerical equivalent 12, (12 x 11 + 8) mod 26 = 140 (mod 26) = 10 = “K”

“a” has the numerical equivalent 0, (0 x 11 + 8) mod 26 = 8 (mod 26) = 8 = “I”

……… and so on …

to give:

Ciphertext:
SKIWS VIJSG VSYKG NASKR GNJIV JJHIV OVGQZ APWA
Cryptanalysis of an affine cipher can be carried out algebraically. If we know that the ciphertext results from an affine cipher, then we know that the encryption equation is of the form 

C  =  (aP  +  b) mod 26

And we know that the decryption equation is therefore of the form

P  =  a-1(C  -  b) mod 26

There is a limited number of possible values for a and b, and therefore a limited number of combinations of a and b that could have been used for the encryption. There are 12 possible values for the multiplier a (the twelve numbers less than 26 that are also relatively prime to 26 – see the list above), and 26 possible values for b, for a total of 12 x 26 = 312 possible combinations of a and b. However, the combination a = 1 b = 0 does nothing, so there are 312 – 1 = 311 nontrivial affine ciphers.

One approach to cracking the cipher, therefore, is to try each pair of values for a and b, substitute in the ciphertext, and see if an intelligible plaintext emerges.

